

Gestión y Modelación de Datos

Sesión Practica: Cursores


MARÍA CONSTANZA PABÓN

MCPABON@JAVERIANACALI.EDU.CO

Cursores

- Una forma eficiente de manejar el resultado de una consulta que retorna un gran conjunto de filas
- Es un apuntador al resultado de la consulta
- Tipos:
 - Implícitos: Oracle PL/SQL crea un cursor implícito cada vez que ejecuta una instrucción SQL
 - Ejemplo: SELECT INTO, INSERT, UPDATE, DELETE
 - Explícitos: declarados y controlados por el programador

Proceso



Ejemplo

Declare

```
CREATE OR REPLACE FUNCTION empleados  
RETURN VARCHAR AS
```

```
  rEmp empleado%ROWTYPE;
```

```
  CURSOR cursEmp IS SELECT empleado.*
```

```
    FROM empleado NATURAL JOIN departamento  
    WHERE nombreDpto = ' Ventas'  
    ORDER BY nombre;
```

```
  correos VARCHAR(500) := '';
```

```
  BEGIN
```

Open

```
    OPEN cursEmp;
```

```
    LOOP
```

Fetch

```
      FETCH cursEmp INTO rEmp;
```

```
      EXIT WHEN cursEmp%NOTFOUND;
```

```
      correos := correos || rEmp.email || ' ; ' ;
```

```
    END LOOP;
```

Close

```
    CLOSE cursEmp;
```

```
    RETURN correos;
```

```
  END;
```

Cursor con parámetros

```
CREATE OR REPLACE PROCEDURE nombresEmp (desdeP NUMBER, hastaP
NUMBER) AS
    CURSOR c1 (desde NUMBER DEFAULT 0,
               hasta NUMBER DEFAULT 10000000)
    IS SELECT nombre FROM empleado
       WHERE salario BETWEEN desde AND hasta;
    nombreV VARCHAR(50);
BEGIN
    OPEN c1 (desdeP, hastaP);
    LOOP
        FETCH c1 INTO nombreV;
        EXIT WHEN c1%NOTFOUND;
        dbms_output.put_line(nombreV);
    END LOOP;
    CLOSE c1;
END;
```

Fetch

- Recupera las filas del conjunto de resultados de una en una
- Cada vez recupera la fila actual (donde está apuntando el cursor) y avanza el cursor a la siguiente fila
- Al hacer fetchs seguidos se recuperan varios registros. Ejemplo:

```
OPEN c1;  
FETCH c1 INTO name1; -- 1er registro  
FETCH c1 INTO name2; -- 2o registro  
FETCH c1 INTO name3; -- 3er registro
```

Atributos del cursor

- Retornan información útil sobre el estado del cursor

Atributo	Descripción
%ISOPEN	Si el cursor está abierto retorna TRUE, de lo contrario, retorna FALSE
%FOUND	Antes del primer FETCH retorna NULL, después, retorna TRUE si el último FETCH recuperó un registro, FALSE de lo contrario
%NOTFOUND	Antes del primer FETCH retorna NULL, después, retorna FALSE si el último FETCH retornó un registro, TRUE de lo contrario
%ROWCOUNT	Antes del primer FETCH retorna 0 (cero), después, retorna el número de registros que se han recuperado

- Uso: nombre_cursor%ISOPEN, nombre_cursor%FOUND, nombre_cursor%NOTFOUND, nombre_cursor%ROWCOUNT

Atributos del cursor

```
DECLARE
    CURSOR c1 IS SELECT nombre FROM empleado ORDER BY nombre;
    nombreV VARCHAR(50);
BEGIN
    IF NOT c1%ISOPEN THEN
        dbms_output.put_line('notOpen');
        OPEN c1;
    END IF;
    LOOP
        FETCH c1 INTO nombreV;
        IF c1%FOUND THEN
            dbms_output.put_line(c1%ROWCOUNT || ': ' || nombreV);
        ELSE EXIT;
        END IF;
    END LOOP;
    CLOSE c1;
END;
```


Recorriendo cursores con FOR

```
CREATE OR REPLACE FUNCTION empleadosFOR
RETURN VARCHAR AS
    rEmp empleado%ROWTYPE;
    CURSOR cursEmp IS SELECT empleado.*
        FROM empleado NATURAL JOIN departamento
        WHERE nombreDpto = ' Ventas' ORDER BY
nombre;
    correos VARCHAR(500) := '';
BEGIN
    FOR rEmp IN cursEmp LOOP
        correos := correos || rEmp.email || ' ';
    END LOOP;
    RETURN correos;
END;
```

El FOR abre el cursor,
hace el fetch en cada
iteración y lo cierra al
terminar de recorrer los
resultados

Cursor for update

- Permite actualizar o borrar el registro en que está posicionado el cursor
- Se debe declarar el cursor FOR UPDATE
 - Si el select del cursor tiene varias tablas, se debe especificar que se va a actualizar o borrar, indicando el atributo (aun para borrar se escribe un atributo)

FOR UPDATE OF tabla.atributo1, tabla.atributo2 ...

- El WHERE del UPDATE o DELETE será “WHERE CURRENT OF”

Cursor for update

```
DECLARE
  rEmp empleado%ROWTYPE;
  CURSOR cursEmp IS SELECT empleado.*
                      FROM empleado NATURAL JOIN departamento
                      WHERE nombreDpto = ' Ventas'
                      FOR UPDATE OF empleado.cc;
BEGIN
  FOR rEmp IN cursEmp LOOP
    IF INSTR(rEmp.email, 'gmail') <> 0 THEN
      DELETE empleado WHERE CURRENT OF cursEmp;
    END IF;
  END LOOP;
END;
```

Variables tipo cursor (REF CURSOR)

- Permite declarar un cursos sin especificar la consulta, ésta se especifica en tiempo de ejecución
- Estas variable se pueden pasar como parámetro o retornar de funciones, para pasar resultados del query entre procedimientos y funciones

Variables tipo cursor (REF CURSOR)

```
DECLARE
    TYPE T_CURSOR IS REF CURSOR;
     cursEmp T_CURSOR;
    rEmp empleado.email%TYPE;   correos VARCHAR(500) := '';
    consulta VARCHAR(50) := 'SELECT email FROM empleado';
BEGIN
    OPEN cursEmp FOR consulta;
    LOOP
        FETCH cursEmp INTO rEmp;
        EXIT WHEN cursEmp%NOTFOUND;
        correos := correos || rEmp || ' ; ' ;
    END LOOP;
    CLOSE cursEmp;
    dbms_output.put_line( correos);
END;
```

Referencias

Manual de Oracle 11g: Database PL/SQL Language Reference.

Cursores.

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/static.htm#LNPLS00602